

Maple 在大学物理教学中的应用

孟 勇

(合肥北辰教育培训学校有限公司, 安徽 合肥 230041)

摘 要 针对大学物理教学中出现的若干问题, 将数学软件 Maple 的四大基本功能用于大学物理各个模块之中。首先在符号演算方面完成了包括动力学习题、亥姆霍兹线圈的磁场分布以及孤立波方程的求解问题。同时在数值计算方面探究了例如有限深方势阱的束缚态能量、椭圆形线圈的电场分布以及在磁场中的弹簧摆运动的问题。此外还通过该软件自带的程序设计功能制作出光栅衍射的效果图、麦克斯韦妖的思想实验程序以及热力学量的计算。最后在图形动画方面中通过画出相轨迹图探究了非线性 RLC 电路的电学量的变化, 并制作出 Alpha 粒子的散射实验的科技动画和氢原子波函数的交互式动态图, 从而展现了 Maple 软件在公式推导、数据可视化、建模仿真等方面辅助大学物理教学的重要作用。

关键词 符号演算; 数值计算; 程序设计; 图形动画

APPLICATION OF MAPLE IN COLLEGE PHYSICS TEACHING

MENG Yong

(Hefei Beichen Education Training School Co. Ltd, Hefei, Anhui 230041)

Abstract Aiming at several problems in college physics teaching, this paper combines the four basic functions of the mathematical software maple to apply it to each module of college physics. Firstly, in the symbolic calculus, the problems including dynamics exercises, the magnetic field distribution of Helmholtz coil and the solution of solitary wave equation are completed. At the same time, the problems such as the bound state energy of the finite deep square potential well, the electric field distribution of the elliptical coil, and the spring pendulum motion in the magnetic field are explored in the numerical calculation. In addition, through the program design function of the software, the effect diagram of grating diffraction, the effect diagram of grating diffraction, the thought experiment program of Maxwell's demon, and the calculation of thermodynamic quantities are produced. Finally, in the aspect of graphic animation, the change of the electrical quantity of the nonlinear RLC circuit is explored by drawing the phase trajectory diagram, and the scientific and technological animation of the scattering experiment of alpha particles and the interactive dynamic diagram of the hydrogen atom wave function are produced. Thus, it demonstrates the important role of Maple software in assisting college physics teaching in formula derivation, data visualization, modeling and simulation.

Key words symbolic calculus; numerical calculation; program design; graphics animation

收稿日期: 2021-09-27

作者简介: 孟勇, 男, 主要从事非线性物理与大学物理教学的研究工作, wanguozhifu@yeah.net。

引文格式: 孟勇. Maple 在大学物理教学中的应用[J]. 物理与工程, 2022, 32(6): 87-103.

Cite this article: MENG Y. Application of Maple in college physics teaching[J]. Physics and Engineering, 2022, 32(6): 87-103. (in Chinese)

大学物理相对于中学物理而言,无论是在深度与广度上都极大地上升了一个层次。这首先体现在中学物理大多数情况下只考虑到运用一些初等数学知识来研究若干均匀变化的物理量。而在大学物理中研究非均匀变化的物理量才是常态。这就必须涉及微积分、线性代数、矢量与张量分析等一系列的高等级的数学工具。因此,对于刚刚进入大学的学生来说如何熟练的运用这些数学知识来解决物理问题变成了学习大学物理的首要难题。同时在中学物理中涉及最多的模型就是点模型,如质点与点电荷模型。所以只涉及到平动,以及简单的摆动、振动。但在大学物理里不仅有像刚体的章动与旋进等复杂的运动,还涉及如氢原子波函数、 δ 势阱等一系列抽象的概念。除此之外,大学物理实验也呈现出“数量多、难度大、意义重”等特点^[1]。这些都无疑增大了学生学习大学物理的难度^[2-4]。而且部分大学生学习目标不明确,学习动力不足,学习思维与方法没有及时从中学物理转化到大学物理中来。以上三点无疑都制约着学生学习大学物理的效果,同时也是考验着大学物理教师的教学水平与能力。

Maple 是由加拿大滑铁卢大学研发面向科研、工程、教学的数学软件^[5-8]。它具有超强性能的符号推演功能、以及无限精度的数值计算能力,同时还有着完整的编程语言与高精度的图形动画技术。因而它在科学与数学领域中享有盛名,被称为“数学家的软件”。并且该软件已经被广泛用于各种领域之中并发挥着重要作用^[9-11]。本文针对 Maple 软件的四大主要功能分别在大学物理各个模块之中的应用,展现出 Maple 在辅助大学物理教学中的作用。

1 符号演算

Maple 软件包含了近 3000 个函数程序库和程序包用来解决微积分、线性代数、概率统计、图论等常见的数学分支问题。因此相比其他数学软件 Maple 的优势就在于其强大的符号计算能力。下面将其符号演算功能应用在大学物理若干实例之中。

1.1 代数运算

在文献^[12]中有一道题目:光滑三棱柱放在

光滑水平桌面上,不可伸长的绳绕过棱脊的轻滑轮,两端各系质量为 m_1 和 m_2 的物体,这两物体分别在棱柱的两个光滑斜面上。求解两物体和棱柱的运动情况。

针对这个问题先绘制出如图 1 的示意图,然后在 Maple 软件平台采用拉格朗日力学方法并主要借助替换命令 subs 与求导命令 diff,以及解方程命令 solve 完成该问题的解答。具体命令见附录 A,运行该代码后得到最终答案。




图 1 模型示意图

$$\ddot{x} = \frac{(m_1 \cos \alpha + m_2 \cos \beta)(m_1 \sin \alpha - m_2 \sin \beta)g}{(m_1 + m_2)(M + m_1 + m_2) - (m_1 \cos \alpha + m_2 \cos \beta)^2} \quad (1)$$

$$\begin{cases} \ddot{x}_1 = \frac{(m_1 \sin \alpha - m_2 \sin \beta)(m_2 \cos \beta - (M + m_2) \cos \alpha)g}{(m_1 + m_2)(M + m_1 + m_2) - (m_1 \cos \alpha + m_2 \cos \beta)^2} \\ \ddot{y}_1 = \frac{(M + m_1 + m_2)(m_1 \sin \alpha - m_2 \sin \beta)g \sin \alpha}{(m_1 + m_2)(M + m_1 + m_2) - (m_1 \cos \alpha + m_2 \cos \beta)^2} \end{cases} \quad (2)$$

$$\begin{cases} \ddot{x}_2 = \frac{((M + m_1) \cos \beta - m_1 \cos \alpha)(m_2 \sin \beta - m_1 \sin \alpha)g}{(m_1 + m_2)(M + m_1 + m_2) - (m_1 \cos \alpha + m_2 \cos \beta)^2} \\ \ddot{y}_2 = \frac{(M + m_1 + m_2)(m_2 \sin \beta - m_1 \sin \alpha)g \sin \beta}{(m_1 + m_2)(M + m_1 + m_2) - (m_1 \cos \alpha + m_2 \cos \beta)^2} \end{cases} \quad (3)$$

从式(1)~式(3)可以看出该问题解的表达式非常复杂。如果通过笔算将会非常耗时耗力,并且还会有一定的出错率。因此,秉承数学机械化的思想^[13]将 Maple 软件融入到教师日常的授课过程中,则可以在实现精准的运算的同时又节省大量时间,从而在保证上课进度的同时将自己的聪明才智与创新能力用于更有意义的教学活动中。

1.2 矢量运算

在电磁学课程中涉及如电场强度的叠加、磁场的求解、洛伦兹力的计算问题时都或多或少需要进行矢量的加减、点乘、叉乘等运算^[14]。但这些计算对刚上大学的新生来说还是有一定的挑战。

如图 2 所示的亥姆霍兹线圈^[15]是由同轴上半径与距离均为 R 的两个线圈组成,并且都是通有电流强度为 I 的同向电流。因此可在 Maple 中通过使用与矢量计算相关的命令,并借助毕奥-萨伐尔定律来计算亥姆霍兹线圈的轴线上磁感应强度大小,通过运行附录 B 中程序后得到结果为

$$B_z = 4i\mu_0 R^2 \left[\frac{1}{(5R^2 - 4zR + 4z^2)^{\frac{3}{2}}} + \frac{1}{(5R^2 + 4zR + 4z^2)^{\frac{3}{2}}} \right] \quad (4)$$

$$\frac{d^2}{dz^2} B_z(0) = 0 \quad (5)$$

$$\frac{d^4}{dz^4} B_z(0) = -\frac{27648i\mu_0 \sqrt{5}}{3125R^5} \quad (6)$$

以及绘制出 z 轴磁感应强度分布图,如图 3 所示。




图 2 亥姆霍兹线圈




图 3 磁感应强度分布

从式(4)可以看出亥姆霍兹线圈在轴线上磁感应强度大小具有很高的对称性,并且由式(5)~

式(6)可知其磁感应强度的在 $z=0$ 处的二阶导数等于 0,同时四阶导数小于 0,这说明 $z=0$ 附件处磁感应强度均匀。并且从图 3 中也可以直观地观察到在两线圈内侧的一定区域磁感应强度大致保持在一个稳定值。这就是亥姆霍兹线圈产生的磁场的主要特征。

1.3 微分方程的求解

大学生在学习大学物理各个模块知识的过程中都或多或少的需要求解各式各样的微分方程,而这些方程的求解对于部分数学基础薄弱的学生来说是一个不小的挑战。因此针对这类学生可借助于 Maple 软件辅助完成对微分方程的求解。此外,部分优秀的大学生在本科阶段就已经开始进行科研活动了,所以他们也需要掌握通过数学软件完成对复杂的微分方程求解的能力。

在文献[16]中介绍了孤子的相关知识,并直接给出描述浅水波运动的 KdV 方程及其双曲函数解,即

$$\frac{\partial y}{\partial t} = \frac{3}{2} \sqrt{g/h} \left(\frac{2}{3} \alpha \frac{\partial y}{\partial x} + y \frac{\partial y}{\partial x} + \frac{1}{3} \sigma \frac{\partial^3 y}{\partial x^3} \right) \quad (7)$$

$$y(x, t) = a \operatorname{sech}^2 \left\{ \frac{1}{2} \sqrt{\frac{a}{\sigma}} \left[x - \sqrt{gh} \left(1 + \frac{a}{2h} \right) t \right] \right\} \quad (8)$$

通过观察式(7)可知其为非线性偏微分方程,其解并不满足叠加原理,所以针对这类方程若直接通过人工求解将十分困难,因此可运用 Maple 软件中的偏微分方程程序包直接进行求解。运行附录 C 所示代码,其结果如图 4 所示。

从图 4(a)可以看出在 Maple 中可以通过 TWSolution 命令直接解得如椭圆函数解、双曲函数解、三角函数等不同类型共 16 个解。而在图 4(b)中通过变换得到的 sech 的双曲函数解 solution1,与此同时从解的表达式发现该孤立波的运动速度与其高度成正比。然后设置相关参数并且在同一幅图中绘制处两个不同高度的孤立波的运动图像。

图 5(a)中,红蓝两色孤立波峰处于同一水平位置,而在图 5(b)~(d)中,随着时间的进行可以明显看出波峰较高蓝色波其运动速度比红色波大,因而两波逐渐远离。这与上述所讨论的孤立波的运动速度与其高度成正比的结论相符合。

```

Solution := TWSolutions(KdV, singso1=false, functions=TWSolutions.functions.allowed)
Solution := {u=2*_C2^2*_C3^2*_JacobiNC(_C3*x+_C4*t+_C2*_C2^2*_C3^2-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4), {u=
=2*_C2^2*_JacobiND(_C3*x+_C4*t+_C2*_C2^2*_C3^2+4*_C2^2*_C3^2-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4), {u=(2*_C2^2*_C3^2
-2*_C2^2)*_JacobiNC(_C3*x+_C4*t+_C2*_C2^2*_C3^2-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4), {u=(-2*_C2^2*_C3^2
+2*_C2^2)*_JacobiND(_C3*x+_C4*t+_C2*_C2^2*_C3^2+4*_C2^2*_C3^2-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4), {u=-2*_JacobiNS(_C3*x
+_C4*t+_C2*_C2^2*_C3^2+4*_C2^2*_C3^2-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4), {u=-2*_JacobiSN(_C3*x+_C4*t+_C2*_C2^2*_C3^2
+_C4*t+_C2*_C2^2*_C3^2+4*_C2^2*_C3^2-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4), {u=-2*_WeierstrassP(_C4*x+_C5*t+_C3*_C2*_C2^2*_C3^2
-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4), {u=-2*_cot(_C2*x+_C3*t+_C2^2*_C3^2-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4), {u=-2*_coth(_C2*x
+_C3*t+_C2^2*_C3^2-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4), {u=-2*_csc(_C2*x+_C3*t+_C2^2*_C3^2-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4), {u=
-2*_csch(_C2*x+_C3*t+_C2^2*_C3^2-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4), {u=-2*_sec(_C2*x+_C3*t+_C2^2*_C3^2-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4),
+4*_C2^2*_C3^2-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4), {u=-2*_tanh(_C2*x+_C3*t+_C2^2*_C3^2-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4),
+_C2^2*_C3^2-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4), {u=-2*_tanh(_C2*x+_C3*t+_C2^2*_C3^2-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4),
+_C2^2*_C3^2-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4)}
N:=nops(Solution);
N:=16
(a)


solution:=Solution[14]
solution:={u=2*_C2^2*_sech(_C2*x+_C3*t+_C1)^2
-4*_C2^2*_C3^2-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4}

solution1:=subs(isolate(4*_C2^2*_C3^2-8*_C2^2*_C3^2-4*_C2^2*_C3^2+_C4=0, _C1=_C2*x[0], _C2
=k, solution)[1]
solution1:=u=2*_k^2*_sech(-4*_k^3*t+k*x+k*x_0)^2
pdetest(solution1, KdV)
0
(b)

```

图4 代码的运行结果

(a) KdV 方程的若干组解析解; (b) KdV 方程的孤立子解

图5 孤立波的运动(红色波: $k=-1, x_0=0$;蓝色波: $k=-1.4, x_0=0$)(a) $t=0$; (b) $t=30$; (c) $t=60$; (d) $t=90$

2 数值计算

在物理学中能精确求解出来的方程毕竟是少数,更多的是解析解根本不存在的方程,因此通过

数值方法求解物理问题显得尤为必要。Maple 除了提供无误差的符号演算功能之外还具有强大的数值计算能力,可以实现无限精度的数值计算。因此可以将 Maple 软件用于大学物理中涉及数值计算问题的求解中。

2.1 超越方程

在文献[17]中有一道题目:一个质量为 μ 的粒子在一维势场 $V(x) = \begin{cases} 0, & |x| > a \\ -V_0, & |x| \leq a \end{cases}$ 中运动,其中 $V_0 > 0$ 。求基态能力 E_0 满足的方程;求存在且仅存在一个束缚态的条件。

随后在该文献的解答中根据波函数及其一阶导数在边界点的连续,列出了以下相关公式

$$E = -|E|, \alpha = \sqrt{\frac{2\mu|E|}{\hbar^2}}, \beta = \sqrt{\frac{2\mu(V_0 - |E|)}{\hbar^2}} \quad (9)$$

$$\eta = \alpha a, \xi = \beta a, Q = \sqrt{\frac{2\mu V_0 a^2}{\hbar^2}} \quad (10)$$

$$\eta = \xi \tan \xi \quad (11)$$

$$\eta = -\xi \cot \xi \quad (12)$$

$$\eta^2 + \xi^2 = Q^2 \quad (13)$$

并巧妙地采用如图6所示数形结合求曲线交点的方法求出存在且仅存在一个束缚态的条件为

$$Q = \sqrt{\frac{2\mu V_0 a^2}{\hbar^2}} < \frac{\pi}{2} \text{ 或 } V_0 a^2 < \frac{\pi^2 \hbar^2}{8\mu} \quad (14)$$




图6 作图法

但是用该方法并不能较为精确地求出该束缚态所对应的能量,因此可以用 Maple 中的 fsolve 命令求解该超越方程的解。其命令如附录 D 所示:最后通过在 Maple 上运行该程序后,其结果如表 1 所示。

在表 1 中不仅能较为精准地得到束缚态下能量 E 具体数值,而且也明显看出随着势阱 V_0 的增大,其束缚态的能量也逐渐降低。以上这些结果与现象是无法用数形结合的方法直观得到的。

表 1 束缚态能量
($\mu=9.1 \times 10^{-31} \text{ kg}, a=0.529 \text{ \AA}$)

V_0/eV	η	ξ	E/eV
1	0.0700643	0.261645	-0.0669938
2	0.134469	0.358681	-0.246766
3	0.194303	0.427021	-0.515229
4	0.250354	0.480407	-0.855362
5	0.303208	0.524310	-1.2546
6	0.353317	0.561578	-1.70361
7	0.401039	0.593917	-2.19489
8	0.446661	0.622439	-2.72268
9	0.490419	0.647915	-3.28228
10	0.532507	0.670900	-3.86983

因此通过 Maple 来求解超越方程确实可以得到更多的较为精确的数值解。

2.2 数值积分

对于均匀带电的圆环产生的电场分布的问题已经在相关文献[18], 文献[19]中有所研究, 而对于均匀带电的椭圆形的导线电场分布问题却少有研究, 因此下面主要通过 Maple 的 int 命令计算得到均匀带电的椭圆形导线的电势数值解。

如图 7 所示: 在 $X-O-Y$ 平面内有一个均匀带电且电荷线密度为 λ 的椭圆形线圈。其半长轴与半短轴长分别是 a 和 b 。因此可以写出该椭圆的参数方程为

$$x(t) = a \cos(t) \quad (15)$$

$$y(t) = b \sin(t) \quad (16)$$




图 7 带电椭圆导线

则线圈上 S 处距离圆形的距离为

$$\rho = \sqrt{a^2 \cos^2 t + b^2 \sin^2 t} \quad (17)$$

并由式(15)~式(16)可知夹角 τ 与参数 t 满足下述关系

$$\tan \tau = \frac{b}{a} \tan t \quad (18)$$

$$d\tau = \frac{ab}{a^2 \cos^2 t + b^2 \sin^2 t} dt \quad (19)$$

则线圈上 S 处的电荷元的电荷为

$$dQ = \lambda \rho d\tau = \frac{\lambda ab}{\sqrt{a^2 \cos^2 t + b^2 \sin^2 t}} dt \quad (20)$$

再由图 8 可知椭圆形线圈在 P 点产生的电势为

$$\varphi = \frac{1}{4\pi\epsilon_0} \int_0^{2\pi} \frac{\lambda ab}{R \sqrt{a^2 \cos^2 t + b^2 \sin^2 t}} dt \quad (21)$$

$$R = \sqrt{(x - a \cos t)^2 + (y - b \sin t)^2 + z^2} \quad (22)$$







图 8 直角坐标系的 P 点坐标

为了方便进行求解设 $\lambda = 4\pi\epsilon_0$, 并在 Maple 上输入附录 E 中命令后, 其结果如图 9 所示。

从图 9(a)和(b)中可以看出椭圆形导线在 x 与 y 轴上的电势分布类似, 图形呈现凹形。都是越靠近导线的位置其电势越高, 同时在线圈内部越靠近中心点的地方电势越低。而通过观察图 9(c)可知在圆环中心点电势最高, 然后往两侧逐渐递减。此外还可以在如图 10 所示的球坐标系下来求解该问题。

则可以通过与在直角坐标系下求解电势的相类似命令进行数值求解得到电势的空间分布图。

从图 11 中可以看出在极角为 $\theta = \pi/2$ 情况下红色线条接近水平, 这是因为处于椭圆形线圈轴中心处极径为 $r = 0.1 \text{ m}$ 的环状区域尺寸远远小于椭圆形线圈, 所以可将整个环状区域视为处于椭圆中心的参考点, 因此其电势基本无变化。而其他的极径曲线在 $\theta = \pi/2$ 和 $\theta = 3\pi/2$ 处都明显出现了峰值。这说明在椭圆短轴方向上出现电势的极大值。此外, 在该图中随着极径 r 的增大, 其曲线峰值逐渐变小, 曲线也趋于平缓。这是由于当极径远远大于椭圆环尺度时, 带电椭圆环可视为一个点电荷, 其在相同距离产生的电势相同。因此在图中 $r = 40 \text{ m}$ 的紫色曲线也趋于水平。

图9 电势沿坐标轴分布($a=4\text{m}$, $b=3\text{m}$)(a) Φ - x 关系; (b) Φ - y 关系; (c) Φ - z 关系图10 球坐标系的 P 点坐标图11 电势在球坐标系下的分布($a=4\text{m}$, $b=3\text{m}$, $\theta=\pi/2$)

2.3 常微分方程

弹簧摆是同时具有振动与摆动两种运动形式的物理模型,在文献[20]中研究了弹簧摆的内共振现象,并数值模拟得到弹簧摆共振时相图、轨迹

图、能量图等。而在文献[21]中通过数值求解的方法考察了在水平面上受稳定约束情况下弹簧摆的运动,发现随着弹簧摆初始能量的增加,其运动情况越来越复杂,但总体上仍保持着一种准周期的运动。

本节将在匀强磁场中探究水平面上带电的弹簧摆的运动。如图12所示:在垂直于光滑水平面的磁感应强度为 B 的匀强磁场中,有一个带电荷量为 q ,并且质量为 m 的绝缘正电小球通过一个原长为 L_0 劲度系数为 k 的轻质弹簧与坐标原点相连。




图12 平面磁力弹簧摆的运动示意图

由于小球受到的弹力始终指向或者背离圆心,所以选择极坐标系下研究该弹簧摆的运动较为合适。因此首先列出极坐标下的牛顿运动学方程

$$m(\ddot{r} - r\dot{\theta}^2) = -k(r - L_0) - Bqr\dot{\theta} \quad (23)$$

$$m(r\ddot{\theta} + 2\dot{r}\dot{\theta}) = Bqr\dot{r} \quad (24)$$

然后通过 Maple 中用 dsolve 命令求出该常微分方程组的数值解,其具体命令如附录 F 所示,通过

运行其代码可得轨迹图如 13 所示。




图 13 弹簧摆的轨迹($m=20\text{g}, k=0.5\text{N/m}, L_0=1, q=0.1\text{C}$)

(a) $B=0, r(0)=1.5\text{m}, \dot{r}(0)=0, \theta(0)=0, \dot{\theta}(0)=1/\text{s}$; (b) $B=1\text{T}, r(0)=1.5\text{m}, \dot{r}(0)=0, \theta(0)=0, \dot{\theta}(0)=1/\text{s}$; (c) $B=1\text{T}, r(0)=1.5\text{m}, \dot{r}(0)=1\text{m/s}, \theta(0)=0, \dot{\theta}(0)=1/\text{s}$; (d) $B=1\text{T}, r(0)=1.5\text{m}, \dot{r}(0)=1\text{m/s}, \theta(0)=0, \dot{\theta}(0)=0$

在图 13(a)中在无磁场的情况下弹簧摆将做准周期运动。而在图 13(b)中则可以明显观察到弹簧摆在磁场中运动轨迹类似于外摆线。同时对图 13(b)~(d)可知,若增大径向速度并且减少横向速度,则弹簧摆的轨迹图像将有尖点出现。

3 程序设计

Maple 具有完善的编程语言,其内核是由 C 语言编译的,所以其语法与 C 语言相似。与此同时它还吸收其他语言的优秀特点,这让 Maple 的使用变得灵活。而且 Maple 本身就含有非常丰富的命令集,所以可以大大降低编程难度与缩短编程所用时间。因此可以将 Maple 编程用于大学物理相关问题的研究之中。

3.1 光栅衍射

衍射光栅通常作用为分光器械,常被用于单色仪与光谱仪。在大学物理中对于光栅衍射实验也作为重要的内容加以学习。如图 14 所示的光栅衍射示意图中:设透射光栅的总缝数为 N ,缝宽




图 14 光栅衍射的示意图

为 a ,缝间不透光部分宽度为 b ,以及衍射角为 ϕ ,则光在 P 点处的光强^[22]为

$$I_{\phi} = I_0 \left(\frac{\sin \alpha}{\alpha} \right)^2 \left(\frac{\sin N\beta}{\sin \beta} \right)^2 \quad (25)$$

其中, $\alpha = \frac{\pi a}{\lambda} \sin \phi, \beta = \frac{\pi(a+b)}{\lambda} \sin \phi$ 且 $\left(\frac{\sin \alpha}{\alpha} \right)^2$ 为

单缝衍射因子, $\left(\frac{\sin N\beta}{\sin \beta} \right)^2$ 为多缝干涉因子。并且

根据光学教材中所述:光强在相邻两个主极大明纹之间有 $N-1$ 个极小值, $N-2$ 个次级大。此外,由于在光栅衍射中多光束干涉受到了单缝衍射调制,所以会导致若干主明纹消失,因此产生缺级现象。而缺级的级数为 k 满足以下关系

$$\begin{cases} (a+b) \sin \theta = k\lambda \\ a \sin \theta = k'\lambda \\ k = \frac{a+b}{a} k' \quad (k' = \pm 1, \pm 2, \pm 3, \dots) \end{cases} \quad (26)$$

对于上述所说的光栅衍射的实验由于受实验者操作水平的影响可能光栅衍射效果不明显。因此可以通过 Maple 软件进行编程,将光栅衍射的实验现象模拟出来。其代码如附录 G 所示,最后通过在 Maple 中输入其命令得到衍射图。

通过观察图 15(a)~(b)可以明显发现在相邻主极大明纹间光强有 4 个极小值和三个次级大,这符合上文中所论述的当光栅总缝数 $N=5$ 的情况。而在图 15(c)~(d)中能够发现第三级主极大明纹消失,这符合上文说的 $k=d/a=3$ 的缺级现象。这些现象都跟上文中理论很好地符合,这说




图 15 光栅衍射的示意图

(a) $\lambda = 632\text{nm}$ 光强分布图; (b) $\lambda = 632\text{nm}$ 条纹分布图; (c) $\lambda = 450\text{nm}$ 光强分布图; (d) $\lambda = 450\text{nm}$ 条纹分布图

明通过 Maple 软件确实可以达到很好的演示实验的作用。

3.2 麦克斯韦妖

熵增原理作为热力学第二定律的一种表述,说明了在孤立体系中熵永远不会自动减少。同时它指明系统演化的方向,甚至可以用熵的增加作为“时间之矢”。但是作为科学研究,其本身就蕴含着突破创新之意,所以它总是驱使着人们去寻找违反热力学第二定律的例子。

麦克斯韦妖是英国著名物理学詹姆斯·克拉克·麦克斯韦为了说明违反热力学第二定律的可能性而在 1871 年提出来的假象的妖怪。这个妖怪可以控制绝热容器里中间挡板上的小门开关,使得这个门选择性地让分子通过,从而人为达到让速度快的分子集中于挡板一边,而让速度慢的分子集中于挡板的另一边。这样不仅使得容器中熵变小,而且人们可以通过容器中产生的温度差来进行热机做功,提供能源。

布里渊在 20 世纪 50 年代曾解释道:由于这只妖怪处于一个密闭、孤立的绝热容器中,妖精将不会看清任何东西,更不用谈区分粒子速度的大小与方向,从而整个系统继续保持均匀、稳定的平衡状态。若从外界提供光能,确实使得这只妖精可

以区分速度大小与方向,然后控制门的开关导致系统自动降低了熵。可是这样有能量输入,整个容器就不再是孤立的绝热体系。因此整个系统不再适用热力学第二定律去描述,也就谈不上与热力学第二定律相违背。

在上文中布里渊从信息论的角度驱逐了这只妖怪,捍卫了热力学第二定律的正确性。但是麦克斯韦妖自从提出之日起,就是物理学界经久不衰、广为传诵的话题,足以说明麦克斯韦妖的无穷魅力。现在继续在 Maple 中通过使用随机数产生函数 rand 来制作出麦克斯韦妖的动画。由于篇幅有限,故列出关键性命令如附录 H 所示。

从其代码可以看出通过两种随机函数的设置产生快慢两种会无规则运动的粒子。然后再配合其他命令进行编程,就能得到麦克斯韦妖的动画,其效果如图 16 所示。




图 16 不同时刻的容器内部

(a) $t=0$; (b) $t=24$; (c) $t=98$; (d) $t=186$

从图 16 可以看出,一开始浅色与深色小球在整个容器中均匀分布,而随着时间的进行,位于挡板下方的快速运动的浅色小球不断地通过小门运动到容器上部分,相反,位于挡板上方的慢速运动的深色小球则不断从容器上方经过小门进入容器下方,从而使得容器变得更有“秩序”,达到熵自动减小的目的。此外,从图中容器的颜色变化可以看出,随着快的浅色小球与慢的深色小球逐渐分

开,容器会自动产生温度差,即容器上方的颜色会逐渐变深变热,而容器下方颜色将逐渐变浅变冷。这些现象都较为符合麦克斯韦妖的特征。

3.3 特性函数与配分函数

特性函数是热力学中的重要概念,它从宏观角度来描述系统的热学性质。由它出发不仅可以通过对它求偏导数等数学运算得到系统的全部热力学函数,而且通过对它进行全微分操作可导出系统热力学的稳定判据。此外,在微观层面上来说配分函数也扮演着类似于特性函数的角色。它包含了粒子体系统计信息的核心,同样对它进行微分等操作可得到众多的热力学宏观量。因此配分函数是连接宏观量与微观量的桥梁。

基于上文所述特性函数与配分函数在研究热力学系统的重要地位后,可在 Maple 进行相关内容的编程,从而达到在输入函数表达式后系统自动求出全部热力学函数的目的。其相应的编程代码如附录 I 所示。将相关代码输入 Maple 中运行可得结果如图 17 所示。

$$\begin{aligned} &CTFP\left(-NkT\left(1+\ln\left(\frac{V}{N}\left(\frac{2\pi mkT}{h^2}\right)^{\frac{3}{2}}\right)\right)\right) \#理想气体 \\ &\rho=-\frac{NkT}{V}, \mu=-kT\left(1+\ln\left(\frac{2V\sqrt{2}\left(\frac{\pi mkT}{h^2}\right)^{\frac{3}{2}}}{N}\right)\right), kT, S=-kN\left(\ln\left(\frac{8V^2\pi^2 m^3 k^3 T^3}{h^6 N^2}\right)+5\right), U=\frac{3}{2}NkT, H \\ &=\frac{5}{2}NkT, G=-NkT\ln\left(\frac{\sqrt{2}}{4}\left(\frac{V^2\pi^2 mkT}{h^2 T^{\frac{5}{2}}}\right)^{\frac{3}{2}}\right), C_v=\frac{3}{2}Nk, H=\frac{5}{2}NkT, C_p=\frac{5}{2}Nk \\ &CTFP\left(-NkT\left(1+\ln\left(8\pi\left(\frac{kT}{hc}\right)^3\right)\right)\right) \#极端相对论气体 \\ &\rho=-\frac{NkT}{V}, \mu=-kT\left(1+\ln\left(\frac{8\pi V k^3 T^3}{h^3 c^3}\right)\right), S=-kN\left(\ln\left(\frac{8\pi V k^3 T^3}{h^3 c^3}\right)+4\right), U=3NkT, H=4NkT, G \\ &=-NkT\ln\left(\frac{h^3 c^3}{8\pi V k^3 T^3}\right), C_v=3Nk, H=4NkT, C_p=4Nk \\ &CTFP\left(-NkT\ln\left(\frac{kT}{hbar\omega}\right)\right) \#正则系综中的谐振子 \\ &\rho=0, \mu=-kT\ln\left(\frac{kT}{h\omega}\right), S=-kN\left(\ln\left(\frac{kT}{h\omega}\right)+1\right), U=NkT, H=NkT, G=-NkT\ln\left(\frac{kT}{h\omega}\right), C_v=kN, H=NkT, C_p \\ &=kN \end{aligned}$$

(a)

$$\begin{aligned} &TF\left(\frac{V^N}{N!h^3}\left(8\pi\left(\frac{1}{\beta hc}\right)^3\right)\right) \#极端相对论气体: \\ &U=3NkT, H=4NkT, F=NkT\ln\left(\frac{N^3 h^3}{8\pi^3 V k^3 T^3}\right)-NkT, G=-NkT\ln\left(\frac{N^3 h^3}{8\pi^3 V k^3 T^3}\right), C_v=3Nk, C_p=4Nk, p \\ &=\frac{NkT}{V}, S=-N\left(\ln\left(\frac{8V k^3 T^3}{N^3 h^3}\right)+4\right), k, \mu=kT\ln\left(\frac{N^3 h^3}{8\pi^3 V k^3 T^3}\right) \\ &TF\left(2\sinh\left(\frac{\beta h\omega}{2}\right)\right) \#N个量子力学谐振子 \\ &U=\frac{N\hbar\omega\cosh\left(\frac{\hbar\omega}{2kT}\right)}{2\sinh\left(\frac{\hbar\omega}{2kT}\right)}, H=\frac{N\hbar\omega\cosh\left(\frac{\hbar\omega}{2kT}\right)}{2\sinh\left(\frac{\hbar\omega}{2kT}\right)}, F=-kT\ln\left(2\sinh\left(\frac{\hbar\omega}{2kT}\right)^{-N}\right), G= \\ &-kT\ln\left(2\sinh\left(\frac{\hbar\omega}{2kT}\right)^{-N}\right), C_v=-\frac{N\hbar^2\omega^2}{4k^2T^2}+\frac{N\hbar^2\omega^2\cosh\left(\frac{\hbar\omega}{2kT}\right)^2}{4\sinh\left(\frac{\hbar\omega}{2kT}\right)^2k^2}, C_p=-\frac{N\hbar^2\omega^2}{4k^2T^2}+\frac{N\hbar^2\omega^2\cosh\left(\frac{\hbar\omega}{2kT}\right)^2}{4\sinh\left(\frac{\hbar\omega}{2kT}\right)^2k^2}, \\ &p=0, S=-\frac{2\sinh\left(\frac{\hbar\omega}{2kT}\right)kT\ln\left(2\sinh\left(\frac{\hbar\omega}{2kT}\right)^{-N}\right)+N\hbar\omega\cosh\left(\frac{\hbar\omega}{2kT}\right)}{2T\sinh\left(\frac{\hbar\omega}{2kT}\right)}, \mu=kT\ln\left(\sinh\left(\frac{\hbar\omega}{2kT}\right)\right) \end{aligned}$$

(b)

图 17 导出全部热力学函数

(a) 特性函数 $F(T, V, N)$; (b) 配分函数 $Z(T, V, N)$

从上图能够观察到当输入特定的特性函数与各种类型的配分函数后就方便、简单、迅速得到如内能、焓、压强、熵、热熔、化学势等热力学物理量。将冗长的运算直接交给电脑来处理,可以把时间节省下来去研究更有意义的物理问题。而且从图(a)与图(b)中对正则系综的运算结果相同更加能够说明算了热力学与统计力学分别从宏观与微观两种角度对热学问题进行研究,即使其研究过程不同,但研究结果却能殊途同归,得到正确的答案。

4 图形动画

Maple 软件包含着较为完善的数据可视化功能,可以制作出高精度的图片与动画。因此可以借助于 Maple 的图形动画技术对大学物理中抽象、难懂而又复杂的物理过程进行具体、形象、直观具体的可视化描述。从而加深对相关物理情景的理解,了解其物理本质。

4.1 非线性 RLC 电路相轨迹图

针对如图 18 所示非线性 RLC 电路,其非线性电容的库伏特性^[23]为

$$u_c = \frac{q}{c} + \epsilon q^x \quad (27)$$




图 18 非线性 RLC 电路

则由基尔霍夫第二定律可得

$$iR + \frac{q}{c} + \epsilon q^x + L \frac{di}{dt} = 0 \quad (28)$$

针对上式考虑 $x=2$ 这种非线性情况,并且为了下文计算方便取 $L=1H$ 和 $C=1F$ 。则由电路电荷与电流关系可将式(28)改写如下平面自治系统

$$\begin{cases} \frac{dq}{dt} = -q - \epsilon q^2 - i \\ \frac{di}{dt} = i \end{cases} \quad (29)$$

令上式等号右边为 0 可得平衡点为 $P_1(0,0)$ 和 $P_2(0,-1/\epsilon)$ 。则该系统在平衡点处的 Jacobi 矩阵分别为

$$M_{P_1} = \begin{bmatrix} -1 & -1 \\ 1 & 0 \end{bmatrix} \quad (30)$$

$$M_{P_2} = \begin{bmatrix} -1 & 1 \\ 1 & 0 \end{bmatrix} \quad (31)$$

则在平衡点 P_1 和 P_2 处可将非线性系统式(29)分别进行线性化为

$$\begin{cases} \frac{di}{dt} = -q - i \\ \frac{dq}{dt} = i \end{cases} \quad (32)$$

$$\begin{cases} \frac{di}{dt} = -i + q \\ \frac{dq}{dt} = i \end{cases} \quad (33)$$

针对式(32)其特征方程与特征根为

$$\begin{vmatrix} -1-\lambda & -1 \\ 1 & -\lambda \end{vmatrix} = \lambda^2 + \lambda + 1$$

$$\lambda_1 = -\frac{1}{2} + \frac{\sqrt{3}}{2}i, \quad \lambda_2 = -\frac{1}{2} - \frac{\sqrt{3}}{2}i \quad (34)$$

由文献[24]可知 λ_1 与 λ_2 为一对共轭复根时,且 $\text{Re}(\lambda_1) < 0$ 时,则原点 P_1 为稳定焦点。同理在式(33)其特征方程与特征根为

$$\begin{vmatrix} -1-\lambda & 1 \\ 1 & -\lambda \end{vmatrix} = \lambda^2 + \lambda - 1$$

$$\lambda_3 = -\frac{1}{2} + \frac{\sqrt{5}}{2}, \quad \lambda_4 = -\frac{1}{2} - \frac{\sqrt{5}}{2} \quad (35)$$

由于 $\lambda_3 \neq \lambda_4$ 都是实数,且 $\lambda_3 \lambda_4 < 0$,则其中坐标点 P_2 为鞍点。下面通过 Maple 中的微分方程画图命令 DEplot 画出其线性系统的相轨迹图,其相关代码见附录 J,运行结果后如图 19 所示。




图 19 线性化系统的相轨迹图

(a) 稳定焦点(0,0); (b) 鞍点(0, -1/ε)

从图 19(a)中逐渐收缩的相轨迹曲线发现平衡点(0,0)是渐进稳定的。这说明此时电阻很大,

系统具有衰减振荡的特征。而在图 19(b)中明显观察到相轨迹曲线先逐渐接近原点然后又远离原点,因此(0, -1/ε)是不稳定平衡点。这也说明了电路会随着时间逐渐失稳。

4.2 Alpha 粒子散射实验动画

卢瑟福的 Alpha 粒子散射实验作为近代物理学中的著名实验,该实验通过用氦原子核轰击薄的金箔发现了 Alpha 粒子大角度散射的现象,进而建立了原子的核式模型。此外,该实验在《原子物理学》课程也作为重点内容加以学习。但是,由于受到客观条件的限制,教师在授课过程中可能会缺乏与之配套的实际演示实验来辅助教学。这样有可能导致部分学生对该实验的无法进行全面性的认识,不利于教学目标的实现。因此可以通过 Maple 中的图形组合命令 display 制作 Alpha 散射实验的仿真动画来直观的描述该过程。其代码如附录 K 所示,运行相关程序可得到 Alpha 粒子散射动画,其效果如图 20 所示。




图 20 Alpha 粒子的散射动画

(a) $t = 5.5 \times 10^{-17} \text{ s}$; (b) $t = 1.4 \times 10^{-15} \text{ s}$; (c) $t = 1.96 \times 10^{-15} \text{ s}$;
(d) $t = 2.53 \times 10^{-15} \text{ s}$

在图 20 中可以看出当大量带正电的 Alpha 粒子以相同的速率从不同位置穿过金原子后,绝大多数粒子将继续保持原来的运动方向运动。其运动轨迹几乎没改变。这说明原子内部存在着较大空间。但同时也能观察到极少数粒子出发了明显的偏转,更少数粒子的运动出现了大角度的偏转。而且从粒子运动留下的深色轨迹线可以发现越接近金原子的粒子其运动的偏转角度越大。甚

至出现了一个粒子被金原子沿原路径反弹回去的情况。这说明金原子内部有一个带正电且质量相对较大的微粒。因此通过 Maple 制作的科技仿真动画确实很好地符合该实验的相关结论。从而可以加深学生对该实验的全面性理解。

4.3 氢原子波函数的交互式图形

量子力学作为构建现代物理大厦的基石之一,是用描述微观世界的物理运动的科学理论。在该理论中可以通过波函数来描述系统的状态,其模长的平方表示粒子在该点出现的概率密度。因此波函数在整个量子力学中举足轻重。

氢原子作为最简单的原子,其波函数是极少数能够精确求解的模型之一。在《量子力学》这门课程中也将其作为中心力场中下求解波函数的典型例子进行学习。并通过分离变量法求出其波函数为

$$\psi(n, l, m) = R_{nl}(r)Y_{lm}(\theta, \phi) \quad (36)$$

其中, $R_{nl}(r)$ 是径向波函数,而 $Y_{lm}(\theta, \phi)$ 是球谐波函数。此外 n, l, m 分别为主量子数、轨道量子数与磁量子数。

由于氢原子的波函数的具体表达式^[25]过于复杂,所以可以通过观察波函数的图像来定性的了解其波函数的性质。因此,下面就利用 Maple 中交互式命令 Explore 制作出该波函数图形。运行附录 L 中程序,可得到结果如图 21~图 23 所示。

在图 21 中通过调节量子数按钮 n 和 l ,从而改变 $|R_{nl}(r)r|^2$ 的图像。从图(a)~(b)对比可知当轨道角动量 l 保持不变时,随着 n 的增大,波函数的峰数逐渐增多,同时图像最右端的主峰也逐渐远离原点且其峰值逐渐降低。相反由图(c)~(d)的可知当保持主量子数 n 不变而增大轨道量子数 l 时,波函数的峰数逐渐减少,其数目为 $n-l$ 。

在图 22 中通过观察图(a)可知 $|Y_{00}(\theta, \varphi)|^2$ 时球形,这说明 s 态中电子云成球状分布的,各个方向上找到电子的概率相同。而在图(b)中可以观察到 $|Y_{10}(\theta, \varphi)|^2$ 成纺锤体,这说明 p 态下不同方向上的概率密度不相等。而再由图(c)与图(d)的对比可知当轨道角动量子数 l 不变时,磁量子数 m 每增大 1 图形就会减少两个“花瓣”。

在图 23 中(a)~(b)的对比中可知:当 $n=7, l=2$ 时, $m=\pm 1$ 的 $|\psi_{nlm}(r, \theta, \varphi)|^2$ 图形是完全相同的,呈四片花瓣状。同时由图(c)与(d)可知当固




图 21 $|R_{nl}(r)r|^2 - r/a$ 的关系





图 22 z 轴方向下的 $|Y_{lm}(\theta, \varphi)|^2$

图 23 $|\psi_{nlm}(r, \theta, \varphi)|^2$ 的空间分布

定主量子数 n 与磁量子数 m 而增大轨道量子数 l , 则图像中的“花瓣”数将随之增多。

通过上述论述可以说明: 与教材中给出的一些固定情况下的氢原子波函数图像相比, 通过在 Maple 软件中通过改变量子数按钮可以方便、清楚而又直观了解波函数在不同情况下的图像。达到所想即所得的目的, 使学生更能深入了解波函数的空间分布特征, 从而达到可视化教学的目的。

5 结语

本文从力、热、电、光、磁及原子物理等众多模块中以若干实例展现出 Maple 软件的具体应用。进而说明了 Maple 软件在辅助大学物理教学的作用。具体来说: 符号演算与数值运算功能缩短了运算时间, 在尽可能保证运算结果的正确性的同时也能进行结果的验算, 进而将人从复杂且具有

重复性的计算中解放出来, 把更多精力用于创造性的思考之中。同时在 Maple 中针对不同问题或者实现某种功能都可以设计出相关程序来解决。而图形动画功能则可以作为人脑的延伸, 弥补空间想象力的不足。并且教师也可以通过生动、形象、直观的图形动画有效地调动学生兴趣与好奇心, 激发他们学习物理的积极性, 提高教学效果。而且这还有助于实现教育技术现代化。除此之外, 熟悉掌握一门专业的数学软件是信息时代下对理工科大学生的要求。因此本文在每一章节内容中针对不同的问题都给出详细的代码, 可供相关人员进行学习使用。

致谢: 谢谢楼森岳教授对笔者的支持与帮助。

附录

A. 求解理论力学习题:

```
eq1:= x1(t)=x(t)-l(t)*cos(alpha)+c[1]; # 物体
m1 的质心的横坐标, 其中 l 为 m1 到滑轮的距离, 而 c[1]
是取决于三棱柱的形状与大小的常数
eq2:= y1(t)=l(t)*sin(alpha); # 物体 m1 的质心纵
坐标
eq3:= x2(t)=x(t)+(L-l(t))*cos(beta)+c[2];
# 物体 m2 的质心横坐标, 其中 L 是绳子的长度, 而 c[2]
同样是取决于三棱柱的形状与大小的常数
eq4:= y2(t)=(L-l(t))*sin(beta); # 物体 m2 的质
心纵坐标
T:=simplify(1/2*M*diff(x(t),t)^2+1/2*m[1]*
(diff(rhs(eq1),t)^2+diff(rhs(eq2),t)^2)+1/2*m[2]*
(diff(rhs(eq3),t)^2+diff(rhs(eq4),t)^2)); # 系统的动
能表达式
V:=-m[1]*g*rhs(eq2)-m[2]*g*rhs(eq4);
# 系统的势能表达式(以滑轮所在高度为零势能参考面)
Lagrange:=collect(simplify(T-V),{diff(l(t),t),
diff(x(t),t)},factor); # 列出拉格朗日量
L1:=subs({diff(l(t),t)=var4,diff(x(t),t)=var2,
l(t)=var3,x(t)=var1},Lagrange); Epr1:=diff(L1,
var2); Epr2:=diff(L1,var1); Epr3:=subs({var1=x(t),
var2=diff(x(t),t),var3=l(t),var4=diff(l(t),t)},
Epr1); Epr4:=subs({var1=x(t),var2=diff(x(t),
t),var3=l(t),var4=diff(l(t),t)},Epr2); Epr5:=
diff(Epr3,t);
Eq1:=simplify(Epr5-Epr4); # 得到第一个拉格朗日
方程;
Epr6:=diff(L1,var4); Epr7:=diff(L1,var3);
```

```
Epr8:= subs({var1 = x(t), var2 = diff(x(t), t), var3 = l(t), var4 = diff(l(t), t)}, Epr6); Epr9:= subs({var1 = x(t), var2 = diff(x(t), t), var3 = l(t), var4 = diff(l(t), t)}, Epr7); Epr10:= diff(Epr8, t);
Eq2:= simplify(Epr10 - Epr9); # 得到第二个拉格朗日方程;
```

```
SOLVE:= simplify(solve({Eq1, Eq2}, {diff(l(t), t, t), diff(x(t), t, t)})); # 求解出绳子和三棱柱的加速度
ALPHA2:= isolate(add(op([1..3], denom(rhs(SOLVE[1]))))= factor(add(op([1..3], denom(rhs(SOLVE[1])))), cos(alpha)^2);
```

```
Solution1:= subs(ALPHA2, SOLVE); # 对三棱柱加速度的表达式进行化简
```

```
Solution2:= simplify(subs(Solution1, {diff(eq1, t, t), diff(eq2, t, t), diff(eq3, t, t), diff(eq4, t, t)})); # 求得并化简出物体  $m_1$  和  $m_2$  加速度的表达式
```

B. 亥姆霍兹线圈轴线磁感应强度的求解:

```
restart; with(LinearAlgebra); with(VectorCalculus); # 清零并导入线性代数与矢量计算的程序包
```

```
r1vec:= <R*cos(phi1) | R*sin(phi1) | -a/2>; # 圆心在  $z=-a/2$  的圆形线圈上源点坐标
```

```
r2vec:= <R*cos(phi2) | R*sin(phi2) | a/2>; # 圆心在  $z=a/2$  的圆形线圈上源点坐标
```

```
r3vec:= <0 | 0 | z>; # z轴上的场点坐标
```

```
r13vec:= r3vec - r1vec; # 计算圆心在  $z=-a/2$  的线圈上源点到轴线上场点的位矢  $\mathbf{r}_{13}$ 
```

```
r23vec:= r3vec - r2vec; # 计算圆心在  $z=a/2$  的线圈上源点到轴线上场点的位矢  $\mathbf{r}_{23}$ 
```

```
e13:= Normalize(r13vec, 2); # 计算  $\mathbf{r}_{13}$  方向上的单位矢量  $\mathbf{e}_{13}$ 
```

```
e23:= Normalize(r23vec, 2); # 计算  $\mathbf{r}_{23}$  方向上的单位矢量  $\mathbf{e}_{23}$ 
```

```
d1:= map(diff, r1vec, phi1); # 位矢  $\mathbf{r}_{13}$  对  $\phi_1$  进行求导
```

```
d2:= map(diff, r2vec, phi2); # 位矢  $\mathbf{r}_{13}$  对  $\phi_1$  进行求导
```

```
Bdir1:= simplify(CrossProduct(d1, e13)); # 计算  $\mathbf{d}_1$  与  $\mathbf{e}_{13}$  的叉乘, 并化简其结果
```

```
Bdir2:= simplify(CrossProduct(d2, e23)); # 计算  $\mathbf{d}_2$  与  $\mathbf{e}_{13}$  的叉乘, 并化简其结果
```

```
BZ:= B[z]= mu[0]*i/(4*Pi)*(int(Bdir1[3]/r13^2, phi=0..2*Pi)+int(Bdir2[3]/r23^2, phi=0..2*Pi)); # 通过积分得到 z 轴方向上的磁感应强度大小
```

```
BzR:= expand(simplify(subs(a=R, BZ))); # 令两线圈距离等于线圈半径 R
```

```
BzRD2:= Diff(B[z](0), z, z)= simplify(subs(z=0, diff(rhs(BzR), z, z))); # 在  $z=0$  位置处磁感应强度对 z 求二阶导数
```

```
BzRD4:= ((Diff(B[z](0), z $ 4)= simplify(subs(z=
```

```
0, diff(rhs(BzR), z $ 4)))) assuming (R>0)); # 在  $z=0$  位置处磁感应强度对 z 求四阶导数
```

```
R:=1; mu[0]:=1; i:=1; # 设置参数
```

```
Fig1:= plot(rhs(BzR), z=-3..3, font=[TIMES, ROMAN, 15], labels=["z", "Bz"], labelfont=[TIMES, ROMAN, 17]);
```

```
Fig2:= textplot({[-1/2, 0.3, "-R/2"], [1/2, 0.3, "R/2"]}, thickness=4, font=["times", "roman", 20]);
```

```
Fig3:= plot([[-1/2, 0], [-1/2, 1]], [[1/2, 0], [1/2, 1]]], font=[TIMES, ROMAN, 15], labels=["z", "Bz"], labelfont=[TIMES, ROMAN, 17], linestyle=3, color=black);
```

```
display(Fig1, Fig2, Fig3);
```

C. KdV 方程的求解:

```
restart; with(PDETools); # 清零并导入偏微分方程程序包
```

```
alias(u=u(x, t)); # 设置 u 为 x 和 t 的二元函数
```

```
KdV:= diff(u, t) + 6*u*diff(u, x) + diff(u, x $ 3) # 写出 KdV 方程的标准形式
```

```
Solution:= TWSolutions(KdV, singsol=false, functions=TWSolutions(functions_allowed)); # 求解出该方程的尽可能多的解
```

```
N:= nops([Solution]); # 计算上述解的个数
```

```
solution:= Solution[14]; # 选取其中 sech 函数的解
```

```
solution1:= subs(isolate(4*_C2^3+_C3=0, _C3), _C1=_C2*x[0], _C2=k, solution)[1]; # 对解的表达式进行整理变换
```

```
pdetest(solution1, KdV); # 检验所得结果的正确性
```

D. 束缚态能级的求解:

```
restart; Digits:=6; # 清零并设置输出量为 6 位精度
```

```
eq1:= Q^2=2*mu*V[0]*a^2/hbar^2; eq2:= mu=0.91*10^(-30), a=0.529*10^(-10), V0=1.602*i*10^(-19); eq3:= hbar=0.6626*10^(-33)/(2*Pi); eq4:= Q^2= subs(eq2, eq3, 2*a^2*mu*V0)/rhs(eq3)^2; eq5:= subs(eq4, eta[i]^2+xi[i]^2=Q^2); eq6:= eta[i]= solve(eq5, eta[i])[1]; eq7:= eta[i]= xi[i]*tan(xi[i]);
```

```
eq8:= seq(fsolve({eq6, eq7}, {eta[i], xi[i]}), i=1..10); # 求解该超越方程的若干解
```

```
eq9:= alpha= sqrt(2*mu*varepsilon/(hbar^2));
```

```
eq10:= isolate(eq9, varepsilon); eq11:= alpha[i]= eta[i]/a; eq12:= seq(subs(eq8, subs(varepsilon=varepsilon[i], alpha=alpha[i], eq11, eq2, eq3, eq10)), i=1..10);
```

```
eq13:= seq(subs(eq12, E[i]=-varepsilon[i]/0.16*10^(-18)*ev), i=1..10); # 得到束缚态的能量
```

E. 均匀带电的椭圆形导线的电势数值解:

```
eq1 := subs(a = 4, b = 3, b * a * sqrt(a^2 * cos(t)^2 +
b^2 * sin(t)^2) / (((a^2 - b^2) * cos(t)^2 + b^2) * sqrt((x
- a * cos(t))^2 + (y - b * sin(t))^2 + z^2))); # 输入
被积函数
```

```
plot([seq([i/10, int(subs(x = i/10, y = 0, z = 0,
eq1), t = 0 .. 2 * Pi, numeric)], i = -100 .. 100)],
style = point, symbol = cross, symbolsize = 15, color =
red, labels = ["x/m", "Φ/V"], labelfont = [TIMES,
ROMAN, 15], font = [TIMES, ROMAN, 15]); # 制作
出 x 轴上电势分部图
```

```
plot([seq([i/10, int(subs(y = i/10, x = 0, z = 0,
eq1), t = 0 .. 2 * Pi, numeric)], i = -100 .. 100)],
style = point, symbol = diagonalcross, symbolsize = 15,
color = blue, labels = ["y/m", "Φ/V"], labelfont =
[TIMES, ROMAN, 15], font = [TIMES, ROMAN,
15]); # 制作出 y 轴上电势分部图
```

```
plot([seq([i/10, int(subs(x = i/10, y = 0, z = 0,
eq1), t = 0 .. 2 * Pi, numeric)], i = -100 .. 100)],
style = point, color = aquamarine, symbol = diamond,
symbolsize = 15, labels = ["z/m", "Φ/V"], labelfont =
[TIMES, ROMAN, 15], font = [TIMES, ROMAN,
15]); # 制作出 z 轴上电势分布图
```

F. 弹簧摆运动方程的数值解:

```
restart;with(plots): # 清零并导入作图程序包
```

```
eq1 := m * (diff(r(t), t, t) - r(t) * diff(theta(t), t)^2) =
-k * (r(t) - L0) - B * q * r(t) * diff(theta(t), t); #
列出径向牛顿运动学方程
```

```
eq2 := m * (r(t) * diff(theta(t), t, t) + 2 * diff(r(t),
t) * diff(theta(t), t)) = B * q * diff(r(t), t); # 列出横
向牛顿运动学方程
```

```
parameter1 := m = 0.02, k = 0.5, L0 = 1, B = 0, q =
0.1; # 设置参数 1
```

```
parameter2 := m = 0.02, k = 0.5, L0 = 1, B = 1, q =
0.1; # 设置参数 2
```

```
Eq1 := subs(parameter1, {eq1, eq2}); # 将参数 1 代入
牛顿方程
```

```
Eq2 := subs(parameter2, {eq1, eq2}); # 将参数 2 代入
牛顿方程
```

```
ini1 := r(0) = 1.5, D(r)(0) = 0, theta(0) = 0, D
(theta)(0) = 1; # 设置初始条件 1
```

```
ini2 := r(0) = 1.5, D(r)(0) = 0.5, theta(0) = 0, D
(theta)(0) = 0; # 设置初始条件 2
```

```
ini3 := r(0) = 1.5, D(r)(0) = 1, theta(0) = 0, D
(theta)(0) = 1; # 设置初始条件 3
```

```
D1 := dsolve({ini1, op(Eq1)}, {r(t), theta(t)}, numeric,
output = listprocedure, method = rkf45, maxfun =
```

```
500000); # 在 Eq1 中代入初始条件 1 进行数值求解
```

```
D2 := dsolve({ini1, op(Eq2)}, {r(t), theta(t)}, numeric,
output = listprocedure, method = rkf45, maxfun =
500000); # 在 Eq2 中代入初始条件 1 进行数值求解
```

```
D3 := dsolve({ini2, op(Eq2)}, {r(t), theta(t)}, numeric,
output = listprocedure, method = rkf45, maxfun =
500000); # 在 Eq2 中代入初始条件 2 进行数值求解
```

```
D4 := dsolve({ini3, op(Eq2)}, {r(t), theta(t)}, numeric,
output = listprocedure, method = rkf45, maxfun =
500000); # 在 Eq2 中代入初始条件 3 进行数值求解
```

```
fig0 := textplot([0.3, 1.7, "y/m"], [1.8, -0.3, "x/
m"], font = [TIMES, ROMAN, 25]); # 添加坐标轴
名称
```

```
fig1:=odeplot(D1, [r(t) * cos(theta(t)), r(t) * sin(theta(t))
], 0 .. 100, numpoints = 20000, labels = ["", ""], view =
[-1.8 .. 1.8, -1.8 .. 1.8], font = [TIMES, ROMAN,
20], color = black); # 绘制 D1 的轨迹图
```

```
fig2:=odeplot(D2, [r(t) * cos(theta(t)), r(t) * sin(theta
(t))], 0 .. 100, numpoints = 10000, labels = ["", ""],
view = [-1.8 .. 1.8, -1.8 .. 1.8], font = [TIMES,
ROMAN, 20], color = red); # 绘制 D2 的轨迹图
```

```
fig3:=odeplot(D3, [r(t) * cos(theta(t)), r(t) * sin(theta
(t))], 0 .. 100, numpoints = 2000, labels = ["", ""],
view = [-1.8 .. 1.8, -1.8 .. 1.8], font = [TIMES,
ROMAN, 20], color = turquoise); # 绘制 D3 的轨迹图
```

```
fig4:=odeplot(D4, [r(t) * cos(theta(t)), r(t) * sin(theta
(t))], 0 .. 100, numpoints = 20000, labels = ["", ""],
view = [-1.8 .. 1.8, -1.8 .. 1.8], font = [TIMES,
ROMAN, 20], color = magenta); # 绘制 D4 的轨迹图
display (fig0, fig1); display (fig0, fig2); display (fig0,
fig3);display(fig0, fig4);
```

G. 光栅衍射仿真程序:

```
restart;with(plots): # 清零并导入作图程序包
```

```
Color := proc(lambda) ; # 编程制作处不同波长的光所
对应的颜色
```

```
if lambda <= 0.435 * 10^(-6) then magenta;
```

```
elif 0.435 * 10^(-6) < lambda and lambda <= 0.450 *
10^(-6) then blue; elif 0.450 * 10^(-6) < lambda and
lambda <= 0.492 * 10^(-6) then cyan;
```

```
elif 0.492 * 10^(-6) < lambda and lambda <= 0.577 *
10^(-6) then green; elif 0.577 * 10^(-6) < lambda and
lambda <= 0.597 * 10^(-6) then yellow;
```

```
elif 0.597 * 10^(-6) < lambda and lambda <= 0.622 *
10^(-6) then orange; elif 0.622 * 10^(-6) < lambda
then red; end if; end proc;
```

```
diffraction := proc(a, d, N, ymax, h, lambda) # 进行
```

光栅衍射光谱图像的编程

```
local alpha, beta, Sin, light_intensity;
Sin := y * (10^-3) / sqrt((y(10^-3))^2 + h^2);
alpha := a * Pi * Sin / lambda; beta := Pi * d * Sin / lambda;
light_intensity := sin(alpha)^2 * sin(N * beta)^2 / (alpha^2 * sin(beta)^2);
print(plot(light_intensity, y = -ymax .. ymax, color = Color(lambda), style = PATCHNOGRID,
numpoints = 200000, axes = boxed, labels = ["y/mm", "I/I_0"], labelfont = [TIMES, ROMAN, 15], font = [TIMES, ROMAN, 15]));
print(plot3d(light_intensity, y = -ymax .. ymax, color = Color(lambda), orientation = [270, 0], style = patchnograd, labels = ["y/mm", "", ""], labelfont = [TIMES, ROMAN, 15], font = [TIMES, ROMAN, 15]));
end proc; # 完成光栅衍射光谱图像的编程
diffraction(5e-6, 10e-6, 5, 80, 1, 632e-9);
diffraction(20e-6, 60e-6, 5, 60, 1, 450e-9);
```

H. 麦克斯韦妖动画的关键性命令:

```
Random := rand(1 .. 10); # 设计随机数为 Random,
并且让其随机数产生范围在 1 到 10 之间
move1 := i -> piecewise(Random(i) <= 5, -1, 1);
# 设置函数名为 move1 的随机数函数,使之随机产生 1 或者 -1
move2 := i -> piecewise(Random(i) <= 5, -2, 2);
# 设置函数名为 move2 的随机数函数,使之随机产生 2 或者 -2.
seq(move1(i), i = 1 .. 30); # 验证 move1 是否随机产生 1 或者 -1
seq(move2(i), i = 1 .. 30); # 验证 move2 是否随机产生 2 或者 -2
FastX || n[i] := FastX || n[i - 1] + move2(FastX || n[i - 1]); # 对于第 i 次循环中的第 n 个速度快的粒子设置它的 X_i 坐标,使之比 X_{i-1} 变化了 2
FastY || n[i] := FastY || n[i - 1] + move2(FastY || n[i - 1]); # 对于第 i 次循环中的第 n 个速度快的粒子设置它的 Y_i 坐标,使之比 Y_{i-1} 变化了 2
SlowX || n[i] := SlowX || n[i - 1] + move1(SlowX || n[i - 1]); # 对于第 i 次循环中的第 n 个速度快的粒子设置它的 X_i 坐标,使之比 X_{i-1} 变化了 1
SlowY || n[i] := SlowY || n[i - 1] + move1(SlowY || n[i - 1]); # 对于第 i 次循环中的第 n 个速度快的粒子设置它的 Y_i 坐标,使之比 Y_{i-1} 变化了 1
```

I. 特性函数与配分函数求解程序:

```
restart; with(PDETools);
interface(showassumed = 0); # 设置假设命令
CTFF := proc(F); # 进行特性函数的编程
local u, P, c, s, f, k, h, g, Mu;
P := -diff(F, V); Mu := diff(F, N);
```

```
s := combine(simplify(-diff(F, T)), ln);
u := simplify(F + T * s); h := u + P * V;
g := combine(simplify(h - T * s), ln); c[v] := diff(u, T); c[p] := diff(h, T);
print(p = P, mu = Mu, S = s, U = u, H = h, G = g, C[v] = c[v], H = h, C[p] = c[p]);
end proc; # 完成特性函数的编程
assume(0 < k, 0 < T, 0 < h, 0 < c, 0 < V, 0 < m, 0 < N, 0 < omega, 0 < beta); # 进行变量假设
TF := proc(Z); # 进行配分函数编程
local u, P, h, s, cv, cp, g, f, Mu;
u := subs(beta = 1/(k * T), -diff(ln(Z), beta));
P := dsubs(beta = 1/(k * T), diff(ln(Z), V)/beta);
s := combine(dsubs(beta = 1/(k * T), ln(1/N!) = -N * (ln(N) - 1), simplify(k * (ln(Z) - beta * diff(ln(Z), beta))))), ln); h := u + P * V;
f := combine(expand(subs(ln(N!) = N * ln(N) - N, simplify(dsubs(beta = 1/(k * T), -k * T * ln(Z))))), ln); g := combine(simplify(h - T * s), ln);
cv := diff(u, T); cp := diff(h, T); Mu := diff(f, N);
print(U = u, H = h, F = f, G = g, C[v] = cv, C[p] = cp, p = P, S = s, mu = Mu); end proc;
CTFF(N * k * T * (1 + ln(V * (2 * Pi * m * k * T / h^2)^(3/2)/N))); # 理想气体
CTFF(-N * k * T * (1 + ln(8 * Pi * V * (k * T / (h * c))^3))); # 极端相对论气体
CTFF(-N * k * T * ln(k * T / (hbar * omega))); # 正则系统中谐振子
TF(V * N * (8 * Pi * (1/(beta * c))^3) * N / (N! * h^(3 * N)));
# 极端相对论气体
TF(2 * sinh(beta * hbar * omega / 2)^(-N)); # N 个量子力学谐振子
TF((2 * Pi / (h * omega * beta))^N); # 正则系统中谐振子
```

J. 非线性 RLC 电路的相轨迹程序:

```
restart; with(plots); with(DETools);
eq1 := diff(i(t), t) = -q(t) - i(t); eq2 := diff(q(t), t) = i(t); eq3 := diff(i(t), t) = -i(t) + q(t); eq4 := diff(q(t), t) = i(t);
initvalues := seq(seq([i(0) = j, q(0) = k], j = -1 .. 1), k = -1 .. 1);
DEplot({eq1, eq2}, [i(t), q(t)], t = -10 .. 10, i = -5 .. 5, q = -5 .. 5, {initvalues}, stepsize = 0.05, arrows = SLIM, color = blue, thickness = 2, linecolor = red, font = [TIMES, ROMAN, 20], labelfont = [TIMES, ROMAN, 30]); # 绘制出关于 P_1 处相轨迹图像
DEplot({eq3, eq4}, [i(t), q(t)], t = -10 .. 10, i = -5 .. 5, q = -5 .. 5, {initvalues}, stepsize = 0.05,
```

arrows = SLIM, color = blue, thickness = 2, linecolor = red, font = [TIMES, ROMAN, 20], labelfont = [TIMES, ROMAN, 30]); # 绘制出关于 P_2 处相轨迹图像

K. Alpha 粒子散射动画程序:

```
restart;with(plots);with(plottools);
eq1 := m * diff(x(t), t, t) = k * Q1 * Q2 * x(t)/(x(t)^2 + y(t)^2)^(3/2); # 列出 Alpha 粒子在 x 轴方向的牛顿定律
eq2 := m * diff(y(t), t, t) = k * Q1 * Q2 * y(t)/(x(t)^2 + y(t)^2)^(3/2); # 列出 Alpha 粒子在 y 轴方向的牛顿定律
m := 4 * 1.6605 * 10^(-27); k := 8.9918 * 10^9; Q1 := 79 * 1.602 * 10^(-19); Q2 := 2 * 1.602 * 10^(-19); # 给相关参数进行赋值
for i from 0 to 20 do; # 设置 alpha 粒子的初始参数
  Z[i] := x(0) = 2 * 10^(-10), D(x)(0) = 6.945392889 * 10^6, y(0) = -72 * 10^(-14) + 9/125000000000000 * i, D(y)(0) = 0;
end do;
for i from 0 to 20 do; # 设置 alpha 粒子的初始参数
  z[i] := x(0) = 2 * 10^(-10), D(x)(0) = -6.945392889 * 10^6, y(0) = -72 * 10^(-12) + 9/12500000000000 * i, D(y)(0) = 0;
end do;
for i from 0 to 20 do; # 进行数值求解
  W[i] := dsolve({eq1, eq2, Z[i]}, {x(t), y(t)}, numeric, output = listprocedure);
end do;
for i from 0 to 20 do; # 进行数值求解
  w[i] := dsolve({eq1, eq2, z[i]}, {x(t), y(t)}, numeric, output = listprocedure);
end do;
ms0 := disk([0, 0], 72 * 10^(-12), color = yellow); # 绘制位于原点的金原子图形
for j from 0 to 100 do; 编程绘制 alpha 粒子散射动画中每一帧的图片
  for i from 0 to 20 do X[i] := rhs(W[i][2](j/10^18)); Y[i] := rhs(W[i][4](j/10^18)); Ms[i] := disk([X[i], Y[i]], 2 * 10^(-12)); end do;
  for i from 0 to 20 do;
    Msl[i] := curve([seq([rhs(W[i][2](k/10^18)), rhs(W[i][4](k/10^18))], k=0..j)], color = red, linestyle = 3, thickness = 1); end do;
  anima[j] := display({seq(Ms[J], J = 0..20), seq(Msl[J], J = 0..20)});
end do; 完成每一帧图形的编程
eq6 := display([seq(anima[j], j = 0..100)], insequence = true, scaling = constrained, axes = box, view = [-2 * 10^(-10) .. 2 * 10^(-10), -2 * 10^(-10) .. 2 * 10^(-10)]); # 将每一帧图片按照顺序组合并依次播放
eq7 := textplot([0.15 * 10^(-9), 0.18 * 10^(-9), "E = 1MeV"], color = black);
display([eq6, eq7, ms0], labels = ["x/m", "y/m"], labelfont = [TIMES, ROMAN, 15], font = [TIMES, ROMAN, 15]);
```

$10^(-10) \dots 2 * 10^(-10), -2 * 10^(-10) \dots 2 * 10^(-10)]$; # 将每一帧图片按照顺序组合并依次播放

```
eq7 := textplot([0.15 * 10^(-9), 0.18 * 10^(-9), "E = 1MeV"], color = black);
display([eq6, eq7, ms0], labels = ["x/m", "y/m"], labelfont = [TIMES, ROMAN, 15], font = [TIMES, ROMAN, 15]);
```

L. 氢原子波函数的交互式图形程序

```
restart;with(plots);with(plottools); P := (l, m, x) -> piecewise(l = 0 and m = 0, (1 - x^2)^(1/2 * m) * (x^2 - 1)^(2 * l * l!), 0 < l and 0 < m, (1 - x^2)^(1/2 * m) * diff((x^2 - 1)^l, x $(1 + m))/(2 * l * l!), 0 < l and m = 0, (1 - x^2)^(1/2 * m) * diff((x^2 - 1)^l, x $(1 + m))/(2 * l * l!)); # 输入连带勒让德函数表达式
Y := (l, m, theta, phi) -> piecewise(0 <= m, (-1)^m * sqrt(1/4 * (2 * l + 1) * (1 - m)! / (Pi * (1 + m)!)) * subs(x = cos(theta), P(l, m, x)) * exp(m * phi * I), m < 0, (-1)^abs(m) * conjugate((-1)^abs(m) * sqrt(1/4 * (2 * l + 1) * (1 - abs(m))! / (Pi * (1 + abs(m))!)) * subs(x = cos(theta), P(l, abs(m), x)) * exp(abs(m) * phi * I))); # 输入球谐函数表达式
altLaguerreL := (n, a, x) -> (-1)^a * n! * LaguerreL(n - a, a, x); R := (n, l, r) -> -2 * sqrt((n - l - 1)! / ((n + 1)! ^ 3) * exp(-r/n) * (2 * r/n)^l * altLaguerreL(n + l, 2 * l + 1, 2 * r/n)/n^2); # 输入径向波函数的表达式
Phi := (n, l, m, r, theta, phi) -> R(n, l, r) * Y(l, m, theta, phi); # 输入氢原子波函数的表达式
Explore(plot(R(n, l, r)^2 * r^2, r = 0..100, color = "NavyBlue", thickness = 3, filled = [color = "Blue", transparency = 0.5], labels = ["r/a", "|R(r)r|^(2)"], labeldirections = [horizontal, vertical], labelfont = [TIMES, ROMAN, 15], font = [TIMES, ROMAN, 15], R(n, l, r)^2 * r^2, r = 0..100, color = "NavyBlue", thickness = 3, filled = [color = "Blue", transparency = 0.5], labels = ["r/a", "|R(r)r|^(2)"], labeldirections = [horizontal, vertical], labelfont = [TIMES, ROMAN, 15], font = [TIMES, ROMAN, 15], view = [0..100, 0..0.10]), n = 1..10, l = 0..9); # 绘制 |R_m(r)r|^2 图像
Explore(plot3d(Y(l, m, theta, phi) * conjugate(Y(l, m, theta, phi)), theta = 0..Pi, phi = 0..2 * Pi, coords = spherical, style = patchnogrid, color = theta, font = [TIMES, ROMAN, 15], orientation = [90, 0, 0], numpoints = 100000, parameters = [l = 0..9, m = -9..9]); # 绘制 |Y_lm(theta, phi)|^2 的图像
Explore(plot3d(Phi(n, l, m, 100, theta, phi) * conjugate(Phi(n, l, m, 100, theta, phi)), theta = 0..Pi, phi =
```

0 .. 2 * Pi, coords = spherical, style = patchngrid, color = theta), parameters = [n = 1 .. 10, l = 0 .. 9, m = -9 .. 9]); # 绘制出 $|\phi_{nlm}(r, \theta, \varphi)|^2$ 的图像

参 考 文 献

- [1] 潘琦. 浅谈大学物理实验教学的特点[J]. 科技资讯, 2010(7): 194.
PAN Q. On the characteristics of college physics experiment teaching[J]. Science & Technology Information, 2010(7): 194. (in Chinese)
- [2] 项阳. “大学物理”课程教学中常见问题探析[J]. 教育教学论坛, 2021(17): 38-41.
XIANG Y. An Analysis of the Common Problems in the Teaching of College Physics[J]. Education On Teaching Forum, 2021(17): 38-41. (in Chinese)
- [3] 陈传文. “大学物理”教学中的若干问题探讨[J]. 教育教学论坛, 2020(19): 347-348.
CHEN C W. Discussion on Some Problems in the Teaching of “College physics”[J]. Education On Teaching Forum, 2020(19): 347-348. (in Chinese)
- [4] 梁秋群. 反思大学物理教学中存在的问题及其对策[J]. 才智, 2018(30): 185-186.
LIANG Q Q. Reflect on the existing problems and countermeasures in college physics teaching[J]. Ability and Wisdom, 2018(30): 185-186. (in Chinese)
- [5] 何青, 王丽芬. MAPLE 教程[M]. 北京: 科学出版社, 2006.
- [6] 张韶华, 王新茂. 符号计算系统 Maple 教程[M]. 合肥: 中国科学技术大学出版社, 2007.
- [7] ROBERT M. Corless. Maple 经典: 科学程序员入门[M]. 北京: 高等教育出版社, 2002.
- [8] 张晓丹, 堵秀凤, 李祥林. Maple 的图形动画技术[M]. 北京: 北京航空航天大学出版社, 2005.
- [9] 鞠磊. Maple 在水文学实验教学中的应用[J]. 高师理科学刊, 2021, 41(3): 102-105.
JU L. Application of Maple in experimental teaching of hydrology[J]. Journal of Science of Teachers' College and University, 2021, 41(3): 102-105. (in Chinese)
- [10] 朱从山. 基于 Maple 软件辅助绘制自由基二元共聚曲线的研究[J]. 高分子通报, 2021(2): 46-53.
ZHU C S. Study on the Drawing of Free Radical Binary Copolymerization Curve Based on Software[J]. Polymer Bulletin, 2021(2): 46-53. (in Chinese)
- [11] 孙玉真. Maple 语言在确立实现企业生产规模中的应用[J]. 现代计算机, 2006(8): 107-110.
SUN Y Z. Application on Production Scale of Enterprises in Maple Language[J]. Modern Computer, 2006(8): 107-110. (in Chinese)
- [12] 梁昆森. 力学. 下册, 理论力学[M]. 4 版. 高等教育出版社, 2009.
- [13] 吴文俊. 数学机械化[M]. 北京: 科学出版社, 2003.
- [14] 丁攀峰. 电磁场课程中矢量分析的教学研究[J]. 教育教学论坛, 2015(15): 136-137.
DING P F. Teaching research of vector analysis in electromagnetic field course[J]. Education On Teaching Forum, 2015(15): 136-137. (in Chinese)
- [15] 薄雪, 李化南. 对亥姆霍兹线圈磁场分布的分析[J]. 大学物理实验, 2021, 34(1): 52-55.
BO X, LI H N. Analysis of Magnetic Field Distribution of Helmholtz Coil[J]. Physics Experiment of College, 2021, 34(1): 52-55. (in Chinese)
- [16] 漆安慎, 杜婵英. 力学: 普通物理学教程[M]. 北京: 高等教育出版社, 1997.
- [17] 陈鄂生. 量子力学习题与解答[M]. 北京: 高等教育出版社, 2012.
- [18] 杨天虎, 李玉宏. 均匀带电细圆环电势和电场的计算[J]. 高师理科学刊, 2020, 40(8): 38-43.
YANG T H, LI Y H. The calculation of the electric potential and electric field of a uniformly charged thin ring[J]. Journal of Science of Teachers' College and University, 2020, 40(8): 38-43. (in Chinese)
- [19] 贾秀敏. 均匀带电圆环片的空间静电场分布[J]. 物理与工程, 2014, 24(2): 54-55+59.
JIA X M. Spatial Electrostatic Field Of Uniformly Charged Ring Slice[J]. Physics and Engineering, 2014, 24(2): 54-55+59. (in Chinese)
- [20] 张义灵, 蓝冬云, 张振美, 等. 基于 Matlab 的弹簧摆内共振与混沌运动研究[J]. 大学物理实验, 2020, 33(4): 84-89.
ZHANG Y L, LAN D Y, ZHANG Z M, et al. Study on Internal Resonance and Chaotic Motion of Spring Pendulum Based on Matlab[J]. Physics Experiment Of College, 2020, 33(4): 84-89. (in Chinese)
- [21] 李阳, 王宏, 韩艳玲, 等. 在水平面上受稳定约束的弹簧振子运动研究[J]. 广西物理, 2015, 36(2): 31-34.
LI Y, WANG H, HAN Y L, et al. Study on the Motion of Spring Vibrator Constrained by Stability in Horizontal Plane[J]. GuangXi Physics, 2015, 36(2): 31-34. (in Chinese)
- [22] 程守洙, 江之永. 普通物理学(下册)[M]. 6 版. 高等教育出版社, 2006.
- [23] 杨红卫, 高冉冉, 彭硕, 等. 非线性电容 LC 电路的位移小参数摄动法分析[J]. 北京工业大学学报, 2019, 45(2): 126-131.
YANG H W, GAO R R, PENG S, et al. Analysis of LC Circuit With Nonlinear Capacitance by the Small Parameter Perturbation Method Based on Displacement[J]. Journal of Beijing University of Technology, 2019, 45(2): 126-131. (in Chinese)
- [24] 贺小明. 常微分方程与动力系统概论[M]. 北京: 北京理工大学出版社, 2010.
- [25] 曾谨言. 量子力学教程[M]. 北京: 科学出版社, 2008.